KaSim reference manual v3

Jérôme Feret and Jean Krivine 1 KappaLanguage.org

June 18, 2012

 $^{^{1}\}mathrm{corresponding}$ author: jean.krivine@pps.univ-paris-diderot.fr





This document is work in progress...

Contents

1	Intr	roduction	7
	1.1	Preamble	7
	1.2	The KaSim engine	7
	1.3	Support	8
2	Inst	allation	11
	2.1	Using precompiled binaries	11
	2.2	Obtaining the sources	11
	2.3	Compilation	12
3	The	command line	13
	3.1	General usage	13
	3.2	Main options	13
	3.3	Advanced options	14
4	The	kappa file	17
	4.1	General remarks	17
	4.2	Agent and token signatures	17
	4.3	Rules	18
		4.3.1 Pure rules	19
		4.3.2 Hybrid rules	21
		4.3.3 Rates	22
		4.3.4 Ambiguous molecularity	22
	4.4	Variables	24
	4.5	Initial conditions	27
5	$\mathbf{A} \mathbf{s}$	imple model	29
	5.1	ABC.ka	29
	5.2	Some runs	30





6	Adv	vanced concepts	33
	6.1	Perturbation language	33
		6.1.1 Adding or deleting agents during a simulation	35
		6.1.2 Using snapshots to define a new initial state	35
		6.1.3 Changing the value of a token	37
		6.1.4 Causality analysis	37
		6.1.5 Flux maps	38
		6.1.6 Updating kinetic rates on the fly	39
	6.2	Link type	40
	6.3	Implicit signature	40
	6.4	Simulation packages	40
	6.5	Simulation parameters configuration	41
		6.5.1 The Influence map	41
7	Free	quently asked questions	45

List of Tables

3.1	Command line: main options	14
3.2	Command line: advanced options	14
4.1	Agent signature expression	18
4.2	Kappa expressions	19
4.3	Example of kinetic rates	23
4.4	Algebraic expressions	24
4.5	Symbol usable in algebraic expressions	25
6.1	Perturbation expressions	34
6.2	User defined parameters	42



Introduction



1.1 Preamble

This manual describes the usage of KaSim, the latest implementation of Kappa, one member of the growing family of rule-based languages. Rule-based modelling has attracted recent attention in developing biological models that are concise, comprehensible, and easily extensible. Although this manual contains a self-contained description of Kappa, it is *not* intended as a tutorial on rule-based modeling.¹

1.2 The KaSim engine

KaSim is an open source stochastic simulator of rule-based models [3, 2, 4] written in Kappa. Basically KaSim takes one or several kappa files as input and generates stochastic trajecto-

¹For an idea of how Kappa is used in a modeling context, the reader might find the following short note useful Agile modelling of cellular signalling (SOS'08). A longer article, expounding on the causal aspects of modeling, is also available: Rule-based modelling of cellular signalling (CONCUR'07), See also this tutorial: Modelling epigenetic information maintenance: a Kappa tutorial (CAV'09).



ries of various observables. KaSim implements Danos and Krivine's simulation algorithm [1] which adapts Gillespie's algorithm [5, 6] to rule-based models.

A simulation event corresponds to the application of a rewriting rule, contained in the kappa files, to the current graph (also called a mixture). The rule is selected according to its activity, ie the number of instances it has in the current mixture, multiplied by its kinetic rate, and applied to one of its possible instances in the graph. It results in a new graph together with an updated activity for all rules (see Fig. 1.1). Importantly, the cost of a simulation event is bounded by a constant that is independent of the size of the graph it is applied to [1].




Figure 1.1: The event loop

Note that KaSim is not equipped with a curve visualization tool. However, data outputs are given in a text format that can be displayed using any standard plotting software such as gnuplot.

1.3 Support

- Kappa language tutorials and downloads: http://kappalanguage.org
- Bug reports should be posted on github: https://github.com/jkrivine/KaSim/issues
- Questions and answers on the kappa-user mailing list: http://groups.google.com/group/kappa-users



- Want to contribute to the project? jean.krivine@pps.univ-paris-diderot.fr



Installation

2.1 Using precompiled binaries

The easiest way to use KaSim is to use pre-compiled versions available at https://github.com/jkrivine/KaSim/downloads. Download the version that corresponds to your operating system (Windows, Linux or Mac OSX) and rename the downloaded file into KaSim. Note that on Mac OSX or linux it might be necessary to give executable permissions to KaSim. This can be done using the shell command: chmod u+x KaSim.

To test whether your program does work, simply type ./KaSim --version on a terminal, from the directory that contains the binaries. If the version is displayed it means that the binaries are indeed compatible with your OS. Otherwise you may need to compile KaSim from the sources (see next Section).

2.2 Obtaining the sources

To obtain KaSim you can either use pre-compiled binaries (see previous section) or compile the sources for your architecture. To do so, download the source code from https://github.com/jkrivine/KaSim and make sure you have a recent ocaml compiler installed (KaSim requires Ocaml 3.12 to compile). From a terminal window type ocamlopt.opt -v. If nothing appears then you need to install Ocaml Native compiler that can be downloaded from http://caml.inria.fr/download.en.html.



2.3 Compilation

Once Ocaml is safely installed, untar KaSim archive and compile following these few steps:

- \$ tar xzvf kasim.tar.gz -d Kappa
- \$ cd Kappa
- \$ make

At the end of these steps you should see, in the Kappa directory, an executable file named KaSim. In order to check the compilation went fine, simply type .\KaSim --version. If the ocaml native compiler ocamlopt.opt is not in the path of your system, you may set the variable OCAMLBINPATH to point to the location of the compiler by editing the corresponding line in the Makefile.

The command line

3.1 General usage

From a terminal window, KaSim can be invoked by typing

```
$ KaSim -i file_1 ... -i file_n [option]
```

where file_i are the input kappa files containing the rules, initial conditions and observables, see Chapter 4 below. Tables 3.1 and 3.2 summarize all the options that can be given to the simulator. Basically, one specifies an upper bound either in bio time (arbitrary time unit), or in number of events. Note that bio-time is computed using Gillespie's formula for time advance (see Fig. 1.1) and should not be confused with CPU-time (it's not even proportional). In doubt, we recommend using a bound in number of events since the cost of one event application is bounded (in CPU time) by a constant, so the simulation time of n events is roughly k times faster than a simulation of k * n events.

3.2 Main options

Table 3.1 summarizes the main options that are accessible through the command line. Options that expects an argument are preceded by a single dash, options that do not need any argument start with a double dash. Note that the option -p specifies the number of points that one wishes to have in the final plot. The interval at which these points will be taken is then computed using the simulation limit defined by the user using option -t or -e. For instance requiring a simulation of 100 points during 10 time units will result in a simulation where observables are recorded every 0.1 time units, while requiring 100 points for 1000 events will result in observable being written every 10 events (be careful that the



event density, ie the number of events per time unit, might vary during a simulation, and thus the two sampling methods can lead to very different repartitions of the moments at which observables are recorded even if they specify the same number of observations).

Table 3.1: Command line: main options

Argument	Description
-e e_{max}	Terminates simulation after $e_{max} \geq 0$ events
$-$ t t_{max}	Terminates simulation after $t_{max} \ge 0.0$ time units
-p n	Produces a data file (default: data.out) with $n \ge 0$ data points
-o file	Set the name of data file to file
-d dir	Output any produced file to the directory dir

3.3 Advanced options

Table 3.2 summarizes the advanced options that are accessible through the command line.

Table 3.2: Command line: advanced options

Argument	Description	
-seed n	Seeds the pseudo-random number generator $n > 0$	
implicit-signature	Automatically deduce agent signatures (see Chapter 6)	
-make-sim sim_file	makes a simulation package out of the input kappa files	
-load-sim sim_file	use simulation package sim_file as input	
gluttony	simulation mode that is memory intensive	
	but that speeds up simulation time	

Example

The command

```
$ KaSim -i model.ka -e 1000000 -p 1000 -o model.out
```

will generate a file model.out containing the trajectories of the observables defined in the kappa file model.ka. The file model.out will contain 1000 data points (ie in this case, a measure will be taken every 1000 events). The command

```
$ KaSim -i init.ka -i rules.ka -i obs.ka -i mod.ka -t 1.5 -p 1000
```



will generate a file data.out (default name) containing 1000 data points of a simulation of 1.5 seconds (arbitrary time units) of the model. Note that the input kappa file is split in 4 files containing, for instance, the initial conditions, init.ka, the rule set, rules.ka, the observables, obs.ka, and the perturbations, pert.ka (see Chapter 4 for details). The order in which the files are given does not matter.



The kappa file

4.1 General remarks

The Kappa File (KF) is the formal representation of your model. We use KF to denote the union of the files that are given as input to KaSim (argument -i). Each line of the KF is interpreted by KaSim as a declaration. If the line is ended by the escape character '\' the continuation of the declaration is parsed onto the next line. Declarations can be: agent and token signatures (Sec. 4.2), rules (Sec. 4.3), variables (Sec. 4.4), initial conditions (Sec. 4.5), perturbations (Sec. 6.1) and parameter configurations (Sec. 6.5). The KF's structure is quite flexible and can be divided in any number of sub-files in which the order of declarations does not matter (to the exception of variable declarations, see Section 4.4 for details). Comments can be used by inserting the marker # that tells KaSim to ignore the rest of the line.

4.2 Agent and token signatures

In Kappa there are two entities that can be used for representing biological elements: agents and tokens. Agents are used to represent complex molecules that may bind to other molecules on specific sites. Tokens are used to represent atomic particles such as calcium ions, atp etc. Tokens cannot bind to each others, they can only appear or disappear. In a given model, agents always have a discrete number of instances while tokens may have a continuous concentration.

In order to use agents or tokens in a model, one needs to declare them first. Agent signatures constitute a form of typing information about the agents that are used in the model. It contains information about the name and number of interaction sites the agent has, and



about their possible internal states. A signature is declared in the KF by the following line:

```
%agent: signature_expression
```

according to the grammar given Table 4.1 where terminal symbol are denoted in (blue) typed font. Symbol Id can be any string generated by regular expression $[a-z \ A-Z][a-z \ A-Z \ 0-9 \ _-+]^*$. Terminal symbol ε stands for the empty symbol.

```
Table 4.1: Agent signature expression  \begin{array}{lll} signature\_expression & ::= & \mathtt{Id}(sig) \\ sig & ::= & \mathtt{Id} \; internal\_state\_list \; , \; sig \mid \varepsilon \\ internal\_state\_list & ::= & \mathtt{Td} \; internal\_state\_list \mid \varepsilon \\ \end{array}
```

For instance the line:

```
%agent: A(x,y~u~p,z~0~1~2) # Signature of agent A
```

will declare an agent A with 3 (interaction) sites x,y and z with the site y possessing two internal states u and p (for instance for the unphosphorylated and phosphorylated forms of y) and the site z having possibly 3 states respectively 0, 1 and 2. Note that internal states values are treated as untyped symbols by KaSim, so choosing a character or an integer as internal state is purely matter of convention.

Token signatures are declared using a statement of the form:

```
%token: ca+ # Signature of calcium token
```

4.3 Rules

Once agents are declared, one may add to the KF the rules that describe their dynamics through time. A pure rule looks like:

```
'my rule' kappa expression 	o kappa expression @ rate
```

where 'my rule' can be any name that will refer to the subsequent rule that can be decomposed into a *left hand side* (LHS) and a *right hand side* (RHS) kappa expressions together with a *kinetic rate expression*. One may also declare a *bi-directional rule* using the convention:

```
'bi-rule' kappa expression \leftrightarrow kappa expression @ rate^+, rate^-
```



Note that the above declaration corresponds to writing, in addition of 'my rule', a backward rule named 'my rule_op' which swaps left hand side and right hand side, and with rate rate⁻.

Kappa and rate expressions are generated by the grammar given in Table 4.2.

If pure rules induce reactions between agents, it is possible to mix agents and tokens in *hybrid rules* (which may also be bi-directional) which look like:

```
kappa expression | token expression \rightarrow kappa expression | token expression @ rate
```

Token expressions are generated by the grammar given Table 4.2 (algebraic expressions are given Table 4.4, but can be replaced for now as any real number).

```
Table 4.2: Kappa expressions
                             agent expression, kappa expression |\varepsilon|
kappa expression
agent expression
                       ::=
                             Id(interface)
                       ::= \varepsilon \mid \text{Id } internal \ state \ link \ state
interface
internal state
                       := \varepsilon \mid \text{`Id}
link state
                             \varepsilon \mid !n \mid !\_ \mid ?
                       ::=
                             algebraic expression: token name
token expression
                       ::=
                             | token expression + token expression
token name
                             Id
                       ::=
                             algebraic expression
rate expression
                       ::=
                             | algebraic expression (algebraic expression)
```

4.3.1 Pure rules

A simple rule

With the signature of A defined in the previous section, the line

```
'A dimerization' A(x), A(y^p) \rightarrow A(x!1), A(y^p!1) @ \gamma
```

denotes a dimerization rule between two instances of agent A provided the second is phosphorylated (say that is here the meaning of p) on site y. Note that the bond between both As is denoted by the identifier !1 which uses an arbitrary integer (!0 would denote the same bond). In Kappa, a bond may connect exactly 2 sites so any occurrence of a bond identifier !n has to be paired with exactly one other sibling in the expression. Note also the fact that



site **z** of **A** is not mentioned in the expression which means that it has no influence on the triggering of this rule. This is the *don't care don't write convention* (DCDW) that plays a key role in resisting combinatorial explosion when writing models.

Adding and deleting agents

Sticking with A's signature, the rule

'budding A' A(z)
$$ightarrow$$
 A(z!1),A(x!1) @ γ

indicates that an agent A free on site z, no matter what its internal state is, may beget a new copy of A bound to it via site x. Note that in the RHS, agent A's interface is not completely described. Following the DCDW convention, KaSim will then assume that the sites that are not mentioned are created in the *default state*, ie they appear free of any bond and their internal state (if any) is the first of the list shown in the signature (here state u for y and 0 for z).

Importantly, KaSim respects the longest prefix convention to determine which agent in the RHS stems from an agent in the LHS. In a word, from a rule of the form $a_1, \ldots, a_n \to b_1, \ldots, b_k$, with a_i s and b_j s being agents, one computes the biggest indices $i \leq n$ such that the agents a_1, \ldots, a_i are pairwise consistent with b_1, \ldots, b_i , ie the a_j s and b_j s have the same name and the same number of sites. In which case we say that the for all $j \leq i$, a_j is preserved by the transition and for all j > i, a_j is deleted by the transition and b_j is created by the transition. This convention allows us to write a deletion rule as:

'deleting A'A(x!1),A(z!1)
$$\rightarrow$$
 A(x) @ γ

which will remove the A agent in the mixture that will match the second occurrence of A in this rule.

Side effects

It may happen that the application of a rule has some *side effects* on agents that are not mentioned explicitly in the rule. Consider for instance the previous rule:

'deleting A'A(x!1),A(z!1)
$$\rightarrow$$
 A(x) @ γ

The A in the graph that is matched to the second occurrence of A in the LHS will be deleted by the rule. As a consequence all its sites will disappear together with the bonds that were pointing to them. For instance, when applied to the graph

$$G = A(x!1,y^p,z^2), A(x!2,y^u,z^0!1), C(t!2)$$



the above rule will result in a new graph $G' = A(x!1,y^p,z^2),C(t)$ where the site t of C is now free as side effect.

Wildcard symbols for link state? (for bound or not), !_ (for bound to someone), may also induce side effects when they are not preserved in the RHS of a rule, as in

'Disconnect A'
$$A(x!_) \rightarrow A(x) @ \gamma$$

or

'Force bind A' A(x?)
$$\rightarrow$$
 A(x!1),C(t!1) @ γ

Both these rule will cause KaSim to raise a warning at rule compilation time.

4.3.2 Hybrid rules

Using KaSim *hybrid rules*, one may declare that an action has effects on the concentration of some particles of the system. For instance a rule may consume atp, calcium ions etc. It would be a waste of memory and time to use discrete agents to represent such particles. Instead one may declare tokens using declarations of the form:

%token: atp

%token: adp

One may then use these tokens in conjunction with a classical rule using the hybrid format:

'hybrid rule'
$$S(x^u!1),K(y!1) \mid 1:atp \rightarrow S(x^p),K(y) \mid 1:adp @ 'k'$$

When applied, the above rule will consume one atp token and produce one adp token. Note that as specified by the grammar given Table 4.2, the number of consumed (and produced) tokens can be given by a sum of the form:

$$lhs \mid a_1:t_1 + \ldots + a_n:t_n \rightarrow rhs \mid a_1':t_1' + \ldots + a_k':t_k' \otimes r$$

where each a_i, a'_i is an arbitrary algebraic expression (see Table 4.4) and each t_i, t'_i is a declared token. In the above hybrid rule, calling n_i, n'_i the evaluation of a_i and a'_i , the concentration of token t_i will decrease from n_i and the concentration of token t'_i will increase from n'_i . Importantly the activity of a hybrid rule like the above one is still defined by | lhs | *r, where | lhs | is the number of embeddings of the lhs of the rule in the mixture, and does not take into account the concentration of the tokens it mentions. As we will see in the next section, it is however possible to make its rate explicitly depend on the concentrations of the tokens using a variable rate.



4.3.3 Rates

As said earlier, Kappa rules are equipped with one or two kinetic rate(s). A rate is a real number, or an algebraic expression evaluated as such, called the *individual-based or stochastic rate constant*, it is the rate at which the corresponding rule is applied per instance of the rule. Its dimension is the inverse of a time $[T^{-1}]$.

The stochastic rate is related to the *concentration-based rate constant* k of the rule of interest by the following relation:

$$k = \gamma (\mathcal{A} V)^{(a-1)} \tag{4.1}$$

where V is the volume where the model is considered, $\mathcal{A} = 6.022 \cdot 10^{23}$ is Avogadro's number, $a \geq 0$ is the arity of the rule (ie 2 for a bimolecular rule).

In a modeling context, the constant k is typically expressed using molars $M := moles l^{-1}$ (or variants thereof such as μM , nM), and seconds or minutes. If we choose molars and seconds, k's unit is $M^{1-a}s^{-1}$, as follows from the relation above.

Concentration-based rates are usually favored for measurements and/or deterministic models, so it is useful to know how to convert them into individual-based ones used by KaSim. Here are typical volumes used in modeling:

- Mammalian cell: $V=2.25\ 10^{-12}l\ (1l=10^{-3}m^3)$, and $\mathcal{A}V=1.35\ 10^{12}$. A concentration of 1M in a mammalian cell volume corresponds to $1.35\ 10^{12}$ molecules; $1nM\approx 1350$ molecules per cell.
- Yeast cell (haploid): $V = 4 \cdot 10^{-14} l$, and $AV = 2.4 \cdot 10^{10}$. A concentration of 1M in a yeast cell volume corresponds to $2.4 \cdot 10^{10}$ molecules; $1nM \approx 24$ molecules per cell. The volume is doubled in a diploid cell.
- E. Coli cell: $V = 10^{-15}l$, and $AV = 10^8$.

A concentration of 1M in a yeast cell volume corresponds to 10^8 molecules; $10nM\approx 1$ molecule per cell.

The table below lists typical ranges for deterministic rate constants and their stochastic counterparts assuming a mammalian cell volume.

4.3.4 Ambiguous molecularity

It is considered malpractice to use a Kappa rule of the form $A(x),B(y) \to ...$ @ γ in a model where this rule could be applied in a context where A and B are sometimes already connected and sometimes disconnected. Indeed, this would lead to an inconsistency in the



	Table	4.3:	Exam	ple	of	kinetic	rates.
--	-------	------	------	-----	----	---------	--------

	1	
process	k	γ
general binding	$10^7 - 10^9$	$10^{-5} - 10^{-3}$
general unbinding	$10^{-3} - 10^{-1}$	$10^{-3} - 10^{-1}$
dephosphorylation	1	1
phosphorylation	0.1	0.1
receptor dimerization	$2 \ 10^6$	$1.6 \ 10^{-6}$
receptor dissociation	$1.6 \ 10^{-1}$	$1.6 \ 10^{-1}$

definition of the kinetic rate γ which should have a volume dependency in the former case and no volume dependency in the latter (see Section 4.3.3).

This sort of ambiguity should be resolved, if possible, by refining the ambiguous rule into cases that are either exclusively unary or binary. Each refinement having a kinetic rate that is consistent with its molecularity. Note that in practice, for models with a large number of agents, it is sufficient to assume that the rule A(x), $B(y) \rightarrow \dots \oslash \gamma$ will have only binary instances. In this case it suffices to consider the approximate model:

'assumed binary AB' A(x),B(y)
$$\to$$
... @ γ_2 'unary AB' A(x,c!1),C(a!1,b!2),B(y,c!2) \to ... @ k_1

There are however systems where even enumerating unary cases becomes impossible. As an alternative, one should use the kappa notation for ambiguous rules:

'my rule'
$$kappa_expression \rightarrow kappa_expression @ \gamma_2(k_1)$$

which will tell KaSim to apply the above rule with a rate γ_2 for binary instances and a rate k_1 for unary instances. The obtained model will behave exactly as a model in which the ambiguous rule has been replaced by unambiguous refinements. However the usage of such rule might slowdown simulation in a significant manner depending on various parameters (such as the presence of large polymers in the model). We give below an example of a model utilizing binary/unary rates for rules¹.

- %agent: A(b,c)
 %agent: B(a,c)
 %agent: C(b,a)
- 4. ##
- 5. %var: 'V' 1
- 6. %var: 'k1' [inf]

¹This model is available in the source repository model/poly.ka.



```
7. %var: 'k2' 1.0E-4/'V'
 8. %var: 'k_off'
                    0.1
 9. ##
10. 'a.b' A(b),B(a) \rightarrow A(b!1),B(a!1) @ 'k2' ('k1')
    'a.c' A(c),C(a) \rightarrow A(c!1),C(a!1) @ 'k2' ('k1')
12. 'b.c' B(c),C(b) -> B(c!1),C(b!1) @ 'k2' ('k1')
13. ##
14. 'a..b' A(b!a.B) -> A(b) @ 'k_off'
15. 'a..c' A(c!a.C) -> A(c) @ 'k_off'
16. 'b..c' B(c!b.C) -> B(c) @ 'k_off'
17. ##
18. %var: 'n' 1000
19. ##
20. %init: 'n' A(),B(),C()
21. %mod: [E] > 10000 do $STOP
22. %def: ''dotSnapshots'' ''true''
```

Notice at lines 10-12 the use of binary/unary notation for rules. As a result binding between between freely floating agents will occur at rate 'k2' while binding between agents that are part of the same complex will occur at rate 'k1'. Line 21 contains a *perturbation* that requires KaSim to stop the simulation after 10,000 events and output the list of molecular species present in the final mixture as a dot file (see Section 6.1) and that we give Figure 4.1.

4.4 Variables

In the KF it is also possible to declare *variables* with the declaration:

```
%var: 'var_name' (algebraic expression or kappa expression)
```

where var_name can be any string and algebraic_expression are defined Table 4.4 (available symbols for variable, constants and operators are given Table 4.5).

For instance the declarations



Table 4.5: Symbol usable in algebraic expressions

variable	Interpretation			
[E]	the total number of simulation events			
	since the beginning of the simulation			
[E+]	the total number of productive events			
[E-]	the total number of null events			
[Emax]	the max (productive) event limit as set by the option -e.			
	Note that if unset $\mathtt{Emax} = \infty$			
[T]	the bio-time of the simulation			
[Tsim]	the cpu-time since the beginning of the simulation			
[Tmax]	the max (bio)-time limit as set by the option -t.			
	Note that if unset Tmax= ∞			
' V '	the value of variable 'v' (declared using the %var: statement)			
t	the concentration of token t			
[inf]	symbol for ∞			
unary/binary_op	Interpretation			
[f]	the intuitive mathematical function or constant			
•	associated to $f \in \{log, sin, cos, tan, sqrt, pi\}$			
[mod]	the modulo operator (infix notation)			
[exp]	the exponentiation operation $x \mapsto e^x$			
[int]	the floor function $x \in \mathbb{R} \mapsto \lfloor x \rfloor \in \mathbb{N}^+$			
+,-,*,/,^	the corresponding mathematical operators (infix notation)			






Figure 4.1: Final mixture obtained after 10,000 events of simulation of the poly.ka model. The infinite rate for cycle closure allows one to obtain a large number of triangles.

%var: 'homodimer' A(x!1),A(x!1)

%var: 'aa' 'homodimer'/2

define 2 variables, the first one tracking the number of embeddings of A(x!1), A(x!1) in the graph over time, while the second divides this value by 2: the number of automorphisms in A(x!1), A(x!1). Note that variables that are used in the expression of another variable must be declared beforehand.

It is also possible to use algebraic expressions as kinetic rates as in

```
%var: 'k_on' 1.0E-6 # per molecule per second 'ab' A(x), A(x) -> A(x!1), A(x!1) @ 'k_on'/2
```

KaSim may output values of variables in the data file (see option -p in Chapter 3) using plot do:

%plot: 'var_name'

One may use the shortcut:

%obs: 'var_name' variable expression

to declare a variable and at the same time require it to be outputted in the data file.



4.5 Initial conditions

The initial mixture to which rules in the KF will be applied are declared as

%init: algebraic expression kappa expression

or

% init: token <- algebraic expression

where algebraic_expression is evaluated before initialization of the simulation (hence all token and kappa expression values in the expression are evaluated to 0). This will add to the initial state of the model mult copies of the graph described by the kappa expression. Again the DCDW convention allows us not to write the complete interface of added agents (the remaining sites will be completed according to the agent's signature). For instance:

%var: 'n' 1000

%init: 'n' (A(),A(y~p))

%init: ca2+ <- 0.39 #mM

will add 1000 instances of A in its default state A(x,y~u,z~0), 1000 instances of A in state A(x,y~p,z~0) and a concentration of 0.39 mM of calcium ions. Recall that the concentration of calcium can be observed during simulation using |ca2+|. As any other declaration, %init can be used multiple times, and agents will add up to the initial state.



A simple model

We describe below the content of a simple Kappa model and give examples of some typical run¹.

5.1 ABC.ka

```
1. #### Signatures
2. %agent: A(x,c) # Declaration of agent A
3. %agent: B(x) # Declaration of B
4. %agent: C(x1~u~p,x2~u~p) # Declaration of C with 2 modifiable sites
5. #### Rules
6. 'a.b'A(x),B(x) \rightarrow A(x!1),B(x!1) @ 'on_rate'#A binds B
7. 'a..b'A(x!1),B(x!1) \rightarrow A(x),B(x) @ 'off_rate'#AB dissociation
8. 'ab.c'A(x!_,c),C(x1~u) ->A(x!_,c!2),C(x1~u!2) @ 'on_rate'#AB binds C
9. 'mod x1'C(x1~u!1),A(c!1) ->C(x1~p),A(c) @ 'mod_rate'#ABC modifies x1
10. 'a.c'A(x,c),C(x1~p,x2~u) -> A(x,c!1),C(x1~p,x2~u!1) @ 'on_rate'#A binds C on x2
11. 'mod x2'A(x,c!1),C(x1~p,x2~u!1) -> A(x,c),C(x1~p,x2~p) @ 'mod_rate'#A modifies x2
12. #### Variables
13. %var: 'on_rate'1.0E-4 # per molecule per second
14. %var: 'off_rate'0.1 # per second
15. %var: 'mod_rate'1 # per second
16. %obs: 'AB'A(x!x.B)
17. %obs: 'Cuu'C(x1~u,x2~u)
```

¹The corresponding kappa file is included in the distribution of KaSim, in the directory models/



```
    18. %obs: 'Cpu'C(x1~p,x2~u)
    19. %obs: 'Cpp'C(x1~p,x2~p)
    20. #### Initial conditions
    21. %init: 1000 A(),B()
    22. %init: 10000 C()
```

Line 1-4 of this KF contains signature declarations. Agents of type C have 2 sites x1 and x2 whose internal state may be u(nphosphorylated) or p(hosphorylated). Recall that the default state of these sites is u (the first one). Line 8, rule 'ab.c' binds an A connected to someone on site x (link type !_) to a C. Note that the only rule that binds an agent to x of A is 'a.b' at line 6. Hence the use of !_ is a commodity and the rule could be replaced by

```
'alt_ab.c' A(x!1,c),B(x!1),C(x1^u) \rightarrow ...
```

There are two main points to notice about this model: A can modify both sites of C once it is bound to them. However, only an A bound to a B can connect on x1 and only a free A can connect on x2. Note also that x2 is available for connection only when x1 is already modified.

5.2 Some runs

We try first a coarse simulation of 100,000 events (10 times the number of agents in the initial system).

```
$ KaSim -i ABC.ka -e 100000 -p 1000 -o abc.out
```

Plotting the content of the abc.out file one notices that nothing of significant interest happen to the observables after 250s. So we can now specify a meaningful time limit by running

```
$ KaSim -i ABC.ka -e 100000 -t 250 -p 1000 -o abc.out
```

which produces the data points whose rendering is given in Fig. 5.1. We will use this model as a running example for the next chapter, in order to illustrate various advanced concepts.






Figure 5.1: Simulation of the ABC model: population of unmodified Cs (observable Cuu in red) drops rapidly and is replaced, in a first step by simply modified Cs (observable Cpu in blue) which are in turn replaced by doubly modified Cs (observable Cpp in red). Note that the population of AB complexes (observable AB in black) stabilizes slightly below 400 individuals after about 20s.



Advanced concepts

6.1 Perturbation language

It is possible to use variables of the model as precondition for triggering a *perturbation* of the simulation. Note that, by default, a perturbation is applied whenever its pre-condition is satisfied and then discarded. Such perturbation is called "one shot". It is however possible to re-apply the same perturbation each time its pre-condition is satisfied and until a certain condition is met, using the repeat ... until constructors.

Basic perturbations are obtained using the declaration:

```
%mod: boolean expression do effect
```

and may be applied repeatedly using:

```
%mod: repeat boolean_expression do effect until boolean_expression
```

where boolean_expression and effect are defined by the grammar given Table 6.1 (the operator rel can be any usual binary relation in $\{<,=,>\}$ and algebraic expressions are defined Table 4.4).

The boolean expression is used as a *precondition* that determines when the perturbation will be triggered, for instance a user writes

```
mod: ([T]>10) && ('v1' /'v2') > 1 do ...
```

to indicate she wishes to trigger a perturbation whenever the simulation time has passed 10 time units and the ratio of variables v1 over v2 is above 1. Recall that the perturbations are "one shot" interventions on the simulation. Possible interventions are described in the following sections using examples.



```
Table 6.1: Perturbation expressions
perturbation expression
                               %mod: perturbation
                               | %mod: repeat perturbation until boolean expression
perturbation
                               boolean expression do effect list
                          ::=
boolean expression
                               algebraic expression rel algebraic expression
                          ::=
                               | (boolean expression | | boolean expression)
                                (boolean expression && boolean expression)
                                [not] boolean expression
                                [true] | [false]
effect list
                               effect; effect_list | effect
                          ::=
effect
                               $ADD algebraic expression agent expression
                                $DEL algebraic expression agent expression
                                t <- algebraic expression
                                 $SNAPSHOT opt filename
                                 $STOP opt_filename
                                 $FLUX opt filename boolean
                                 $TRACK 'var_name' boolean
                                 $UPDATE 'var_name' algebraic expression
opt filename
                          ::=
                               ''filename'' \mid arepsilon
boolean
                               [true] | [false]
                          ::=
```



6.1.1 Adding or deleting agents during a simulation

Continuing with the ABC model, the perturbation effect:

$$ADD n C(x1^p)$$

will add $n \ge 0$ instances of C with x1 already in state p (and the rest of its interface in the default state as specified line 4 of ABC.ka). Also the perturbation effect:

will remove all Bs connected to some agent from the mixture.

There are various ways one can use perturbations to study more deeply a given kappa model. A basic illustration is the use of a simple perturbation to let a system equilibrate before starting a real simulation. For instance, as can be seen from the curve given in Fig. 5.1, the number of AB complexes is arbitrarily set to 0 in the initial state (all As are disconnected from Bs in the initial mixture). In order to avoid this, one can modify the kappa file the following way: we set the initial concentration of C to 0 by deleting line 22. Now we introduce Cs after 25 t.u using the perturbation:

The modified kappa file is available in the source repository, in the model/ directory (file abc-pert.ka). Running again a simulation (a bit longer) by entering in the command line:

\$ KaSim -i ABC-pert.ka -e 100000 -t 300 -o abc2.out one obtains the curve given in Fig. 6.1.

6.1.2 Using snapshots to define a new initial state

In the previous example, we let the system evolve for some time without its main reactant C in order to let other reactants go to a less arbitrary initial state. One may object that this way of proceeding is CPU-time consuming if one has to do this at each simulation. An alternative is to use the \$SNAPSHOT primitive that allows a user to export a snapshot of the mixture at a given time point as a new (piece of) kappa file. For instance, the declaration:

will ask KaSim to export the mixture the first time the percentage of null events reaches 90%. The exported file will be named prefix_n.ka where n is the event number at which the snapshot was taken. Note that one may omit to define a prefix and simply type:






Figure 6.1: Simulation of the ABC model with a perturbation: for t<25s, only 'a.b' and 'a.b' rules may apply. This enables the concentration of 'AB' complexes to go to steady state, before introducing fresh Cs at t=25s.



%mod: [E-]/[E]>0.9 do \$SNAPSHOT

in which case the default prefix snap will be used for naming snapshots.

Note that if the name already exists a counter will be appended at the end of the file to prevent overwriting. Note that snapshots can be performed multiple times, for instance every 1000 events, using the declaration:

```
%mod: repeat ([E+] [mod] 1000)=0 do $SNAPSHOT "abc" until [false]
```

which results in KaSim producing a snapshot every 1000 (productive) events until the simulation ends. The perturbation \$STOP "final_state" will terminate the simulation whenever its precondition is satisfied and produce a snapshot of the last mixture. Note that instead of producing kappa files, one may use snapshot perturbations to produce an image of the mixture in the dot format using the parameter "dotSnapshots" (see Section 6.5).

6.1.3 Changing the value of a token

The concentration of any token can be reset on the fly using a perturbation. For instance the declaration:

mod: repeat (|a|<100 do a <- |a|*2) until [false]

will double the concentration of token a each time it gets below 100.

6.1.4 Causality analysis

In our ABC example, adding the instruction:

%mod: [true] do \$TRACK 'Cpp' [true]

will asking KaSim to turn on causality analysis for the observable 'Cpp' since the beginning of the simulation, and display the causal explanation of every new occurrence of 'Cpp', until the end of the simulation. The explanation, that we call a *causal flow*, is a set of rule application ordered by causality and displayed as a graph using dot format. In this graph, an edge $r \longrightarrow r'$ between two rule applications r and 'r'' indicates that the first rule application has used, in the simulation, some sites that were modified by the application of the former. We show Fig. 6.2 an example of such causal flow.

Note that causal flow can show or hide more causal dependencies using various compression techniques that can be enabled as a simulation parameter (see Section 6.5). Causality analysis of the observable Cpp can be turned off at any time using a declaration of the form:






Figure 6.2: Causal flow for the observable 'Cpp' of the ABC model. Plain arrows represent causal dependency, dotted arrows show asymmetric conflict between rule occurrences. Here the 'ab.c' rule has to occur before the 'a.b' rule. Red observable indicate that the last rule allowed one to observe a new instance of 'Cpp'.

%mod: [T]>25 do \$TRACK 'Cpp' [false]

Causal flows are currently in beta testing. This section will be expanded later.

6.1.5 Flux maps

The flux map is a powerful observation that tracks, on the fly, the influence that rule applications have on each others. It is dynamically generated and tracks effective impacts (positive or negative) a every rule application. The flux map can be computed using declarations of the form:

%mod: [true] do \$FLUX "flux.dot" [true]
%mod: [T]>20 do \$FLUX "flux.dot" [false]

The resulting flux map is a graph where a positive edge between rules r and s (in green) indicates an overall positive contribution of r over s. Said otherwise, the sum of r applications increased the activity of s. Conversely, a negative edge (in red) will indicate that r had an overall negative impact on the activity of s. Note that the importance of the flux between two rules can be observed by looking at the label on the edges that indicate the overall activity transfer (positive or negative) between the rules. The above declaration produce a flux map that is shown Fig. 6.3. Note that flux may vary during time, therefore the time






Figure 6.3: Flux map of the abc.ka model, taken from t=0 to t=20 time units. The A releasing rules a..b and mod x2 are contributing very little to the activity of a.c which is a sign of an excess of free As in the system at this time interval.

or event limit of the simulation is of importance and will likely change the aspect of the produced map.

6.1.6 Updating kinetic rates on the fly

Any variable between simple quotes can be updated during a simulation using a declaration of the form:

This perturbation will be applied whenever the observable 'Cpp' will become greater than 500. Its effect will be to set the on rate of all binding rules to 0. Note that according to the grammar given Table 6.1, one may use any algebraic expression as the new value of the variable. For instance:



will cause the on rate of all rules to decrease a hunderd fold. Note that it is possible to override the kinetic rate of a specific rule: in our ABC example, the declaration:

will set the kinetic rate of rule 'a.b' to infinity.

6.2 Link type

In standard kappa, in order to require a site to be bound for an interaction to occur, one may use the *semi-link* construct !_ which does not specify who the partner of the bond is. For instance in the variable:

will count the number of As and Bs connected to someone, including the limit case A(x!1), B(y!1). It is sometimes convenient to specify the type of the semi-link, in order to restrict the choice of the binding partner. For instance the variable:

will count the number of As whose site x is connected to a site y of B, plus the number of Bs whose site y is connected to a site x of A. Note that this still includes the case A(x!1), B(y!1).

6.3 Implicit signature

KaSim permits users in a hurry to avoid writing agent signatures explicitly using the option --implicit-signature of the command line. The signature is then deduced using information gathered in the KF. Note that it is not recommended to use the DCDW convention for introduced agents in conjunction with the --implicit-signature option unless the default state of all sites is mentioned in the %init declarations or in the rules that create agents.

6.4 Simulation packages

The simulation algorithm that is implemented in KaSim requires an initialization phase whose complexity is proportional to R*G where R is the cardinal of the rule set and G the



size of the initial mixture. Thus for large systems, initialization may take a while. Whenever a user wishes to run several simulations of the *same* kappa model, it is possible to skip this initialization phase by creating a *simulation package*. For instance:

KaSim -i abc.ka -t
$$n$$
 -make-sim abc.kasim

will generate a standard simulation of the abc.ka model, but in addition, will create the simulation package abc.kasim (.kasim extension is not mandatory). This package is a binary file, ie not human readable, that can be used as input of a new simulation using the command:

```
KaSim -load-sim abc.kasim -t k
```

Note that this simulation is now run for k time units instead of n. Importantly, simulation packages can only be given as input to the same KaSim that produced it. As a consequence, recompiling the code, or obtaining different binaries, will cause the simulation package to become useless.

6.5 Simulation parameters configuration

In the KF (usually in a dedicated file) one may use expressions of the form:

```
%def: "parameter name" "parameter value"
```

where tunable parameters are described table 6.2 (default values are given first in the possible values column).

6.5.1 The Influence map

The influence map of a model is an object that may help modelers checking the consistency of the rule set they use. It is generated if the modeler has required it using the paramter "dumpInfluenceMap" which will generate the so called *influence map* of the model.

Unlike the flux map, the influence map is statically computed and does not depend on kinetic rates nor initial conditions. It describes how rules with no side effect may potentially influence each other during a simulation. KaSim will produce a dot format file containing the influence relation over all rules and observables of the model. The produced graph visualized using a circular rendering¹ is given in Figure 6.4. Observables are represented as circular nodes² and rules as rectangular nodes. Edges are decorated with the list of embeddings

¹One may use for instance the circo program that is part of the *graphviz* suite.

²Observable can be viewed as identity rules for computing influence.

6.5. SIMULATION PARAMETERS CONFIGURATION

Table 6.2: User defined parameters		
parameter	possible values	description
Causality analysis		
"displayCompression"	any combination of	
	"none", "strong", "weak"	type of compression
"cflowFileName"	"cflow.dot", any string	file name for causal flows
Pretty printing		
"plotSepChar"	" " or any character	separation character for plots
"dotSnapshots"	"false", "true"	generate dot snapshots
"colorDot"	"false", "true"	use colors in dot format files
"progressBarSymbol"	"#" or any character	symbol for the progress bar
"progressBarSize"	"60" or any integer	length of the progress bar
Simulation options		
"dumpIfDeadlocked"	"false","true	Snapshot when simulation
•		is stalled
"maxConsecutiveClash"	"2" or any integer	number of consecutive clashes
	, G	before giving up
		square approximation
Miscellaneous		
"dumpInfluenceMap"	"false","true"	Output the influence map
-		of the model
"influenceMapFileName"	"im.dot", any string	file name for the influence map

(separated by a semi-colon) allowing rules's right hand sides to be mapped to left hand sides. Note that the influence map will not display relations between rules that are induced by side effects (see Section 4.3.1).

More formally, consider the rules $r: L \to R$ and $s: L' \to R'$. One wishes to know whether it is possible that the application of rule r over a graph G creates a new instance of rule s. To do so, KaSim will try to generate a term T containing a match for R and L' that overlap on some sites that are modified by r (see Figure 6.5 for illustration).






Figure 6.4: The influence map of the abc.ka model defined in Chapter 5. Note that observable $C(x1^u, x2^u)$ is not activated by any other rule. This implies that its number of occurrences does not change during a simulation. Edge labels denote embeddings with the convention that $[i \to j]$ denotes the embedding of agent number i of the origin's rhs, into agent j of the target's lhs.






Figure 6.5: Computation of the influence of the top rule on the rule below: the right hand side of the first rules embeds in a common term with the left hand side of the second rule. It results that the first rule has a positive influence on the second.

Chapter 7

Frequently asked questions

Simulation hangs after a while

If the progress bar seems stalled, it does not necessarily mean that the simulation is blocked. In particular when a simulation is triggered with a *time* limit (-t option of the command line) it might only indicate that the bio clock is stalled while computation events still occur. Recall that the average (bio) time one has to wait in order to apply a rule is 1/A, where A is the sum of all the rule activities (which is equal to the number of instances that a rule has, times its kinetic rate). Whenever the number of occurrences of a rule grows too fast (if new agents are created during the simulation for instance), or if the kinetic rate of a rule is defined by a function that grows rapidly, the average time increment might tend to 0 and if it remains so for a while, it will block the progress bar whose advance is proportional to the bio time [T].

In order to make sure that KaSim is not incorrectly blocked you may wish to plot the event clock against time clock using the observable <code>%obs: 'events'[E]</code> or run the simulation using an event limit (-e option of the command line) instead of a time limit.

Perturbation is never triggered or triggered too late

Make sure that perturbation's precondition can be satisfied by KaSim. In particular use inequalities rather than equalities, especially when the precondition is depending on time. Also some perturbations require to be persistent to observe their effect. In this case make sure you used the do ... until construction.



What do null events mean, why do I have any?

Null events is a way for KaSim to compensate for some over approximation it is doing, in order deal with large simulations more efficiently. They do not impact significantly the performances of the simulator. The ratio r of null event over productive ones (that you can track using the observable %obs: 'r'[E-]/[E+] should tend to 0 when models have a lot of agents.

No data points are generated

Make sure you have **%obs** or **%plot** instructions in your KF. Also make sure to use the -p option in the command line to tell KaSim how many points you wish to have on your curves.

Too many instances of an observable

The value of a kappa expression E is equal to the number of embeddings it has in the current mixture M. Embeddings are maps from agents in E to agents in E. If E has symmetries then every permutation of E will be counted as a new embedding. For instance let E = A(x!1), A(x!1) and let $M = A(x!1,y^p)$, $A(x!1,y^u)$. KaSim will count two instances of E in E: the one mapping the first E of E to the first E of E to the second E of E.

The computed influence map is incorrect, it misses some activation or has to much of them

The influence map contains relations that are computed on side effect free rules only. It is likely that a missing activation is due to a side effect that is not taken into account. If the influence map shows an activation between rule r and s that is never possible with a given model, just remember that activation computation implies that there exists a context in which applying rule r will create a new instance of rule s. This context might simply never be realized with the given rules or initial conditions.



Value nan in the data file at the end of the simulation

The value nan means "Not a Number". It is generated when a plotted variable is infinite. Make sure this variable is not divided by zero at some point.



Bibliography

- Vincent Danos, Jérôme Féret, Walter Fontana, and Jean Krivine. Scalable simulation of cellular signaling networks. In *Proc. APLAS'07*, volume 4807 of *LNCS*, pages 139–157, 2007.
- [2] Vincent Danos, Jérôme Feret, Walter Fontanta, Russ Harmer, and Jean Krivine. Rule based modeling of biological signaling. In Luís Caires and Vasco Thudichum Vasconcelos, editors, *Proceedings of CONCUR 2007*, volume 4703 of *LNCS*, pages 17–41. Springer, 2007.
- [3] Vincent Danos and Cosimo Laneve. Formal molecular biology. *Theoretical Computer Science*, 325, 2004.
- [4] James R. Faeder, Mickael L. Blinov, and William S. Hlavacek. Rule based modeling of biochemical networks. *Complexity*, pages 22–41, 2005.
- [5] Daniel T. Gillespie. A general method for numerically simulating the stochastic time evolution of coupled chemical reactions. *Journal of Computational Physics*, 22(4):403– 434, 1976.
- [6] Daniel T. Gillespie. Exact stochastic simulation of coupled chemical reactions. *Journal of Physical Chemistry*, 81(25):2340–2361, 1977.